

Глава 15. IF(), SWITCH(), BLANK() и другие DAX-функции условий

Это продолжение перевода книги Роб Колли. Формулы DAX для Power Pivot. Главы не являются независимыми, поэтому рекомендую начать сначала.

[Предыдущая глава](#) [Содержание](#) Следующая глава

Добавим условную логику в наши DAX-формулы. Рассмотрим рост по сравнению с предыдущим годом (Year-Over-Year, YOY) из [предыдущей главы](#):

[Pct Sales Growth YOY] =
([Total Sales] - [Total Sales DATEADD 1 Year Back]) / [Total Sales DATEADD 1 Year Back]

Значение за 2001 год возвращает ошибку, потому что продажи в прошлом году [Total Sales DATEADD 1 Year Back] равны 0. Это действительно ошибка деления на 0.

Pct Sales Growth YOY	
2001	#ЧИСЛО!
2002	99,9 %
2003	49,9 %
2004	104,8 %
Общий итог	101,5 %

Рис. 15.1. Ошибка #Число! для 2001 года: деление на ноль

Ситуацию легко исправить с помощью интуитивно понятной функции IF():

[Pct Sales Growth YOY] =
IF(
 [Total Sales DATEADD 1 Year Back] <> 0;
 ([Total Sales] - [Total Sales DATEADD 1 Year Back]) / [Total Sales DATEADD 1 Year Back];
 0
)

Pct Sales Growth YOY	
2001	0,0%
2002	99,9%
2003	49,9%
2004	104,8%
Общий итог	101,5%

Рис. 15.2. Теперь мера возвращает 0% вместо ошибки

Функция BLANK()

Мы можем еще улучшить отображение. 0% означает, что у нас был нулевой рост, тогда как на самом деле этот расчет вообще не имеет смысла для 2001 года. Поэтому вместо 0 мы можем вернуть функцию BLANK():

[Pct Sales Growth YOY] =
IF(
 [Total Sales DATEADD 1 Year Back] = 0;
 BLANK();
 ([Total Sales] - [Total Sales DATEADD 1 Year Back]) / [Total Sales DATEADD 1 Year Back]
)

Pct Sales Growth YOY	
2002	99,9%
2003	49,9%
2004	104,8%
Общий итог	101,5%

Рис. 15.3. Теперь 2001 год просто не отображается

Это очень полезный трюк. Запомните такое использование функции BLANK()!

Если мы добавим в сводную таблицу второе поле *Значений*, которое для 2001 года возвращает значение, отличное от нуля, строка за 2001 год вновь проявится:

Pct Sales Growth YOY	Total Sales
2001	\$3 266 374
2002	99,9% \$6 530 344
2003	49,9% \$9 791 060
2004	104,8% \$9 770 900
Общий итог	101,5% \$29 358 677

Рис. 15.4. 2001-й год будет отображаться, если хотя бы одна мера возвращает непустой результат

И наоборот, если вам не нравится отсутствие значений в каком-либо поле, вы можете настроить сводную, и отражать там нечто иное. Для этого активируйте сводную таблицу, перейдите на вкладку *Анализ*, и в левой части ленты кликните *Параметры* (или кликните правой кнопкой мыши на сводной и выберите *Параметры сводной таблицы*). Установите значения, которые следует отражать вместо ошибок и пропусков:

Параметры сводной таблицы

Имя сводной таблицы: PivotTable7

Макет и формат | **Итоги и фильтры** | Вывод | Печать | Данные | Замещающий текст

Макет

Объединить и выровнять по центру ячейки с подписями

Отступ для названий строк в сжатой форме: 1 зн.

Отображать поля в области фильтра отчета: вниз, затем поперек

Число полей фильтра отчета в столбце: 0

Формат

Для ошибок отображать: 0

Для пустых ячеек отображать: 0

Автоматически изменять ширину столбцов при обновлении

Сохранять форматирование ячеек при обновлении

OK Отмена

Рис. 15.5. Настройка сводной таблицы для отражения ошибочных и пустых значений

Функция DIVIDE()

DIVIDE(<числитель>; <знаменатель> [; <значение, если знаменатель равен нулю>])

DIVIDE() – безопасное деление; третий аргумент является необязательным, и по умолчанию возвращает BLANK(), что удобно в большинстве случаев. Можно записать:

```
[Pct Sales Growth YOY using DIVIDE] =  
DIVIDE(  
    [Total Sales] - [Total Sales DATEADD 1 Year Back];  
    [Total Sales DATEADD 1 Year Back]  
)
```

Элегантно, не правда ли?

Шаблон IF...THEN...ELSE полезен не только при делении, и пригодится во многих иных ситуациях. Для деления же мы рекомендуем использовать DIVIDE(), и забыть про проблему с нулевым знаменателем.

Шаблон IF(<test>; <DAX выражение>; BLANK()) мы также рекомендуем запомнить.

Функция ISBLANK

В Excel есть функция ЕПУСТО(). Она возвращает значение ИСТИНА, если ячейка пуста, и значение ЛОЖЬ в иных ситуациях. Функция нужна, когда мы хотим отличить пустую ячейку от ячейки, содержащей пробелы, или пустой текстовой строки "=", или строки с формулой, возвращающей пустое/нулевое значение.

Если обратиться к примеру выше, функция ISBLANK() может быть полезна, если мы хотим отличить ситуацию, когда [Total Sales DATEADD 1 Year Back] возвращает законный 0 (то есть, строки продаж имелись, но сумма столбца SalesAmount = 0; редко, но возможно). Поэтому большую часть времени мы просто проверяем "=0". Но когда вы хотите отличить 0 от BLANK(), ISBLANK() – это то, что вам нужно.

HASONEVALUE()

Excel позволяет стандартными средствами показать в сводной таблице долю от суммы по столбцу. А вот долю в каждой категории – нет. Создадим следующую меру:

```
[Subcategory pct of Category Sales] =  
[Total Sales] /  
CALCULATE(  
    [Total Sales];  
    ALL(Products[SubCategory])  
)
```

И поместим ее в сводную таблицу вместе с [Total Sales]:

Названия строк	Total Sales	Subcat pct of Cat Sales
Accessories	\$700 760	100,00%
Bike Racks	\$39 360	5,62%
Bike Stands	\$39 591	5,65%
Bottles and Cages	\$56 798	8,11%
Cleaners	\$7 219	1,03%
Fenders	\$46 620	6,65%
Helmets	\$225 336	32,16%
Hydration Packs	\$40 308	5,75%
Tires and Tubes	\$245 529	35,04%
Bikes	\$28 318 145	100,00%
Mountain Bikes	\$9 952 760	35,15%
Road Bikes	\$14 520 584	51,28%
Touring Bikes	\$3 844 801	13,58%
Clothing	\$339 773	100,00%
Caps	\$19 688	5,79%
Gloves	\$35 021	10,31%
Jerseys	\$172 951	50,90%
Shorts	\$71 320	20,99%
Socks	\$5 106	1,50%
Vests	\$35 687	10,50%
Общий итог	\$29 358 677	100,00%

Рис. 15.6. Продажи и их доля в категории

Возможно, вы сочтете, что все показанные значения 100,0% подытогов и общих итогов бесполезны. Чтобы подавить их можно использовать функцию HASONEVALUE(). Она проверяет, содержит ли контекст фильтра одно наименование (и возвращает ИСТИНА) или несколько, т.е., относится к промежуточным/общим итогам (и возвращает ЛОЖЬ).

```
[Subcategory pct of Category Sales] =
IF(
    HASONEVALUE(Products[SubCategory]);
    [Total Sales] / CALCULATE(
        [Total Sales];
        ALL(Products[SubCategory])
    );
    BLANK ()
)
```

Названия строк	Total Sales	Subcat pct of Cat Sales
Accessories	\$700 760	
Bike Racks	\$39 360	5,62%
Bike Stands	\$39 591	5,65%
Bottles and Cages	\$56 798	8,11%
Cleaners	\$7 219	1,03%
Fenders	\$46 620	6,65%
Helmets	\$225 336	32,16%
Hydration Packs	\$40 308	5,75%
Tires and Tubes	\$245 529	35,04%
Bikes	\$28 318 145	
Mountain Bikes	\$9 952 760	35,15%
Road Bikes	\$14 520 584	51,28%
Touring Bikes	\$3 844 801	13,58%
Clothing	\$339 773	
Caps	\$19 688	5,79%
Gloves	\$35 021	10,31%
Jerseys	\$172 951	50,90%
Shorts	\$71 320	20,99%
Socks	\$5 106	1,50%
Vests	\$35 687	10,50%
Общий итог	\$29 358 677	

Рис. 15.7. Промежуточные и общие итоги для меры [Subcat pct of Cat Sales] подавлены HASONEVALUE() эквивалентно IF(COUNTROWS(VALUE())=1.

Мы могли бы отключить промежуточные и/или общие итоги на вкладке *Конструктор* на ленте. Но это также отключит подытоги/итоги для столбца [Total Sales]. А мы хотели сделать это только для столбца [Subcat pct of Cat Sales].

IF() на основе полей сводной таблицы

Выше мы использовали IF() для сравнения со значением меры. Но что, если мы хотим проверить, где мы «находимся» с точки зрения контекста фильтра? Например, мы хотим вычислить что-то немного по-другому для конкретной страны? Добавим в модель таблицу подстановки SalesTerritory. Она содержит столбец Country, который мы поместим в строки сводной таблицы для меры Продажи родителям [Sales to Parents]

Sales to Parents = CALCULATE([Total Sales]; Customers[NumberChildrenAtHome] > 0)

Названия строк	Sales to Parents
Australia	\$2 486 889
Canada	\$762 530
France	\$1 400 775
Germany	\$1 460 348
United Kingdom	\$1 666 415
United States	\$3 655 420
Общий итог	\$11 432 377

Рис. 15.8. Сумма для Канады не вызывает доверия

Допустим, у нас есть сомнения в данных по столбцу [NumberOfChildren]. Возможно, то, как мы собираем данные в Канаде, делает это число ненадежным. А этот столбец лежит в основе расчета меры [Sales to Parents]. Поэтому для Канады, и только для Канады, мы хотим заменить эту меру другой мерой, [Sales to Married Couples]. Поэтому мы введем новую меру:


```
[Sales to Parents Adj for Canada] =
IF(
    HASONEVALUE(SalesTerritory[Country]);
    IF(
        VALUES(SalesTerritory[Country]) = "Canada";
        [Sales to Married Couples];
        [Sales to Parents]
    );
    BLANK()
)
```

Функция VALUES() возвращает контекст фильтра, установленный в сводной таблице. Если вы находите в обычной ячейке, он возвращает одно значение. Если вы в строке подытогов/итогов, он возвращает несколько значений:

Названия строк	Sales to Parents	Названия строк	Sales to Parents
Australia	\$2 486 889	Australia	\$2 486 889
Canada	\$762 530	Canada	\$762 530
France a	\$1 400 775	France b	\$1 400 775
Germany	\$1 460 348	Germany	\$1 460 348
United Kingdom	\$1 666 415	United Kingdom	\$1 666 415
United States	\$3 655 420	United States	\$3 655 420
Общий итог	\$11 432 377	Общий итог	\$11 432 377

Рис. 15.9. Формула VALUES(SalesTerritory[Country]) вернет для (a) значение Canada; для (b) – Australia, Canada, France, Germany, United Kingdom, United States

Мы не можем непосредственно проверить условие IF(SalesTerritory[Country]). Оно нарушает [правило для мер](#) – не использовать голые столбцы в качестве аргументов мер. Ранее мы «заворачивали» столбец в какую-нибудь арифметическую функцию. Но, поскольку Country – это текстовая строка, нам нужно использовать какую-то иную функцию. Вот почему мы выбрали VALUES(). В итоге условие проверки выглядит так: IF(VALUES(SalesTerritory[Country])="Canada").

Если мы выполним тест IF(VALUES(...))="Canada" в ячейке, которая вернет более одного значения, мы получим ошибку. Поэтому нам нужно предварительно проверить, что мы находимся в ячейке с единственным значением страны. Для этого мы используем конструкцию IF(HASONEVALUE(SalesTerritory[Country]))

Вот что мы в итоге получим:

Названия строк	Sales to Parents	Sales to Parents Adj for Canada
Australia	\$2 486 889	\$2 486 889
Canada	\$762 530	\$1 078 215
France	\$1 400 775	\$1 400 775
Germany	\$1 460 348	\$1 460 348
United Kingdom	\$1 666 415	\$1 666 415
United States	\$3 655 420	\$3 655 420
Общий итог	\$11 432 377	

Рис. 15.10. Две меры отличается только для Канады

Мы также убрали итоговое значение из столбца Sales to Parents Adj for Canada. Оно было бы не корректным, так как значение для Канады мы взяли из другого столбца.

Значение, возвращаемое VALUES(), не зависит от того, выводится ли поле в сводную таблицу

Посмотрите на сводную таблицу, которая показывает количество продуктов по категориям и цветовой гамме:

	A	B	C
1			
2		Названия строк ▾	Product Count
3		▣ Accessories	35
4		Black	3
5		Blue	3
6		Grey	1
7		NA	24
8		Red	3
9		Silver	1
10		▣ Bikes	125
11		Black	43
12		Blue	13
13		Red	27
14		Silver	19
15		Yellow	23
16		▣ Clothing	48
17		Black	19
18		Blue	3
19		Multi	18
20		White	4
21		Yellow	4
22		▣ Components	189
23		Black	64
24		Blue	9
25		NA	32
26		Red	33
27		Silver	25
28		Silver/Black	7
29		Yellow	19
30		Общий итог	397

Рис. 15.11. Что именно возвращает мера VALUES(Products[Color]) для выделенной ячейки?

Для выделенной на рис. 15.11 ячейки, мера VALUES(Products[Color]) возвращает количество продуктов всех цветов: {"Black", "Blue", "Red", "Silver", "Yellow"}. Обратите внимание, что "Grey" и "NA" не возвращаются для этой ячейки. Но эти два цвета возвращаются для категории Accessories. Это связано с тем, что Category и Color (поля в строках) являются столбцами таблицы Products, что означает, что фильтр категорий влияет на допустимость цвета. Категория велосипеды "Bikes" фильтрует таблицу Products, а велосипеды в цветовой гамме "Grey" или "NA" отсутствуют.

Если мы удалим поле Color из сводной таблицы, что вернет мера VALUES(Products[Color])?

	A	B	C
1			
2		Названия строк ▾	Product Count
3		Accessories	35
4		Bikes	125
5		Clothing	48
6		Components	189
7		Общий итог	397

Рис. 15.12. Итоговые значения по категориям не изменятся

Независимо от того, было ли поле Color выведено в сводную или нет, выделенная нами ячейка C10 на рис. 15.11 не имела никаких фильтров, связанных с цветом.

VALUES() может возвращать уникальные значения

Мы использовали меру [Product Count] для определения числа продуктов в категории (всего или в той или иной цветовой гамме). Например (см. рис. 15.11), имеется 35 продуктов в категории Accessories, принадлежащих к 6 различным цветам. Для подсчета числа цветов в категории используем меру

[Color Values] = COUNTROWS(VALUES(Products[Color]))

	A	B	C
1			
2		Названия строк	Color Values
3		Accessories	6
4		Bikes	5
5		Clothing	5
6		Components	7
7		Общий итог	10

Рис. 15.13. Мера [Color Values] возвращает только уникальные значения для цвета

Характерно, что итоговая строка на рис. 15.13 возвращает значение 10, а не сумму цветов по категориям. Если убрать фильтр со столбца Category, разных цветов у всех продуктов будет 10.

SWITCH()

Если мы хотим проверить несколько условий, вложенные IF() – один из способов решения, но функция SWITCH() намного проще и прозрачнее. В модели данных мы добавили вычисляемый столбец, чтобы увидеть работу функции SWITCH() в действии, но ее можно использовать и в мерах (рис. 15.14).

```
=
SWITCH (
  [Country];
  "United States"; "North America";
  "Canada"; "North America";
  "France"; "Europe";
  "Germany"; "Europe";
  "United Kingdom"; "Europe";
  "Rest of the World"
)
```

SalesTerritoryRegion	SalesTerritoryGroup	Continent
Northwest	North America	North America
Northeast	North America	North America
Central	North America	North America
Southwest	North America	North America

ar | Exch Rates | MinListPrice | PriceTiers | SalesTerritory

Рис. 15.14. Функции SWITCH() определяет континент на основе страны

Вот как работает SWITCH(). Первый аргумент – проверяемое выражение. Начиная со второго аргументы SWITCH() работают парами: нечетный аргумент сравнивается с проверяемым выражением и в случае совпадения возвращает следующий за ним четный аргумент. Например, если проверяемое значение "United States", то возвращается "North America", если "France" –

возвращается "Europe". Если вы заканчиваете SWITCH() "четным" аргументом, он рассматривается как "ELSE" и работает без пары. Т.е., если среди четных аргументов {"United States"; "Canada"; ...; "United Kingdom"}, не найдется проверяемое выражение, то функция SWITCH() вернет "Rest of the World".

SWITCH TRUE()

Функция SWITCH() может быть еще более универсальной, когда в качестве первого аргумента используется TRUE(). Далее можно указать условия для сопоставления, а не точные значения. Например, добавим еще один вычисляемый столбец в таблицу Products для указания диапазона цен прайс-листа:

```
=  
SWITCH (  
  TRUE ();  
  [ListPrice] < 100; "$";  
  [ListPrice] < 500; "$$";  
  [ListPrice] < 1000; "$$$";  
  "$$$$"  
)
```

category	ListPriceBucket
lothing	\$
lothing	\$
lothing	\$
omponents	\$\$
omponents	\$\$
ikes	\$\$\$
ikes	\$\$\$
ikes	\$\$\$

les Products Calendar Exch Rates MinL

Рис. 15.15. Конструкция SWITCH TRUE() позволяет указать условия для проверки

Здесь, как и раньше, начиная со второго, аргументы SWITCH() работают парами. Однако вместо соответствия определенному значению они оценивают условие; если оно истинно, то выбирается парное значение. Например, если значение [List Price] < 100, то возвращается "\$". Если вы заканчиваете SWITCH() "четным" аргументом, это рассматривается как "ELSE". В этом случае "\$\$\$\$" возвращается, если ни одно из условий не совпадает. Обратите внимание, что важен порядок условий: если первое из них выполняется, дальнейших проверок не будет. Поэтому, если вы указали [List Price] < 1000 в качестве первого условия, оно будет верно, и для изделия по цене \$90, и \$400.

Для ускорения работы модели данных лучше создавать вычисляемые столбцы за пределами Power Pivot (например, в Excel, базе данных, или с помощью Power Query), а затем импортировать их как часть исходной таблицы.